

Burroughs Corporation



COMPUTER SYSTEMS GROUP
SANTA BARBARA PLANT

P.S. 2201 6752
B1800/B1700 SORT LANGUAGE

PRODUCT SPECIFICATION

REV LTR	REVISION ISSUE DATE	APPROVED BY	REVISIONS
I	6-11-79	<i>Male</i>	Changes for Mark 9.0 Release
			7-1 Changed SORT.FILES - BIT(24) to BIT(8) Added FILLER and SORT.PURGE.INPUT.FILES
J	7		?

"THE INFORMATION CONTAINED IN THIS DOCUMENT IS CONFIDENTIAL AND PROPRIETARY TO BURROUGHS CORPORATION AND IS NOT TO BE DISCLOSED TO ANYONE OUTSIDE OF BURROUGHS CORPORATION WITHOUT THE PRIOR WRITTEN RELEASE FROM THE PATENT DIVISION OF BURROUGHS CORPORATION"

TABLE OF CONTENTS

GENERAL	1-1
RELATED DOCUMENTATION	1-2
FILE STATEMENT	2-1
SYNTAX	2-1
SEMANTICS	2-3
SORT STATEMENT	3-1
SYNTAX	3-1
SEMANTICS	3-2
USER OPTION LIST	4-1
SYNTAX	4-1
SEMANTICS	4-4
FILE NAMES	4-10
SORT COMMUNICATE	5-1
REINSTATE STATUS	5-1
SORT EOJ COMMUNICATE	6-1
SORT INFORMATION TABLE FORMAT	7-1
SORT KEY TABLE	8-1
SORT EXECUTION	9-1
SAMPLE EXECUTIONS	9-1
ERROR MESSAGES	10-1

GENERAL

The SORT program accepts SORT language input containing file statements, a SORT statement, and user option statement(s). It generates the communicate required to execute the SORT function of the Master Control Program (MCP). The SORT program causes a designated file to be sorted, using tape or disk, on a series of specified keys in ascending or descending order, as required.

The sort language specifications are read from a file labelled CARDS or from the SPO depending on Program Switch 0. If SW0=1, input will be read from the SPO (ACCEPT). The length of the input record scanned for statements is 160 bytes for SPO or MCP default for the file CARDS. This may be altered by modifying the file CARDS for record size and no default.

The workfile medium depends on the resources available and/or the medium requested by the user. There are no default I/O media for this program. INPLACE SORT or TAPESORT should be specified, for example, when minimal disk resources are available for sorting. Note that the scratch workfile for a DISKSORT (vector sort) will be approximately $2 \frac{1}{3}$ times the size of the input file. In contrast, the scratch workfile for an INPLACE SORT is the same size as the input file.

SORT statements are in free-field format and are made up of words, reserved words, numbers, symbols, and reserved symbols. The statements may be permuted in any legitimate order. SORT LANGUAGE input is defined, in BNF, as:

```
<SORT LANGUAGE INPUT> ::= <FILE STATEMENTS><SORT STATEMENT>  
                           <USER OPTION STATEMENT>
```

Comments are allowed between statements and not inside. The entire input record is scanned for text. For those who wish to use sequence numbers in columns 73 to 80 of their input, syntax errors may be prevented by either putting a colon (:) in column 72 or modifying the file CARDS to NOT DEFAULT and RECORD.SIZE = 72. Input from data recorder devices must be delimited by the colon (:), as the record size cannot be modified.

BURROUGHS CORPORATION
COMPUTER SYSTEMS GROUP
SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
B1800/B1700 SORT LANGUAGE
P.S. 2201 6752 REV. J

RELATED DOCUMENTATION

<u>NAME</u> ----	<u>NUMBER</u> -----
B1800/1700 TAPE SORT	P.S. 2212 5181
B1800/1700 DISK SORT	P.S. 2212 5264
B1800/1700 INPLACE SORT	P.S. 2212 5538
B1800/1700 MERGE	P.S. 2212 5207
B1800/1700 SDL (BNF Version)	P.S. 2212 5405
B1800/1700 Software Operational Guide	#1068731

BURROUGHS CORPORATION
 COMPUTER SYSTEMS GROUP
 SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
 B1800/B1700 SORT LANGUAGE
 P.S. 2201 6752 REV. J

FILE STATEMENT

SYNTAX

```

<FILE STATEMENT> ::=      FILE <INPUT FILE DESCRIPTION>
                           <OUTPUT FILE DESCRIPTION>
                           | FILE <MULTIPLE INPUT FILE DESCRIPTIONS>

<MULTIPLE INPUT FILE
DESCRIPTIONS> ::=          <1 TO 8 <INPUT FILE DESCRIPTION>>

<INPUT FILE
DESCRIPTION> ::=          IN <FILE-NAME><INPUT FILE PARAMETERS>
                           | IN <FILE-NAME><DISK IN DEFAULT OPTION>

<FILE-NAME> ::=          <MULTI-FILE-ID>
                           | <MULTI-FILE-ID>/<FILE-ID>
                           | <PACK-ID>/<MULTIFILE-ID>/
                           | <PACK-ID>/<MULTI-FILE-ID>/<FILE-ID>
                           | <EMPTY>

<FILE-ID> ::=            <LABEL>

<MULTI-FILE-ID> ::=      <LABEL>

<PACK-ID> ::=            <LABEL>

<LABEL> ::=              <ALPHANUMERIC CHARACTER STRING>
                           | "<ALL CHARACTERS EXCEPT QUOTE>"

<INPUT FILE
PARAMETERS> ::=          (<INPUT MEDIA><RECORD SIZE>
                           <RECORDS PER BLOCK> <INPUT OPTION LIST>)

<INPUT MEDIA> ::=        CARD | CARDS | PAPER | TAPE
                           | DISK (<RECORDS PER AREA> [<NUMBER OF AREAS>])

<RECORDS
PER AREA> ::=            <UNSIGNED INTEGER>

<NUMBER OF AREAS> ::=    <UNSIGNED INTEGER>

```

BURROUGHS CORPORATION
 COMPUTER SYSTEMS GROUP
 SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
 B1800/B1700 SORT LANGUAGE
 P.S. 2201 6752 REV. J

```

<RECORD SIZE> ::=          <UNSIGNED INTEGER>

<RECORDS
PER BLOCK> ::=          <UNSIGNED INTEGER> | <EMPTY>

<INPUT OPTION
LIST> ::=          <INPUT OPTION>
| <INPUT OPTIONS> <INPUT OPTION LIST>

<INPUT OPTION> ::=          <PURGE OPTION>
| <VARIABLE OPTION>
| <MULTI OPTION>
| <OPTIONAL OPTION>

<PURGE OPTION> ::=          PURGE | <EMPTY>

<VARIABLE OPTION> ::=          VARIABLE | V(<MAX BLOCK SIZE>)
| <EMPTY>

<MAX BLOCK SIZE> ::=          <UNSIGNED INTEGER>

<MULTI OPTION> ::=          MULTI | <EMPTY>

<OPTIONAL OPTION> ::=          OPTIONAL | <EMPTY>

<DISK IN
DEFAULT OPTION> ::=          (DISK <PURGE OPTION> DEFAULT)

<OUTPUT FILE
DESCRIPTION> ::=          OUT <FILE-NAME> <OUTPUT FILE PARAMETERS>
| OUT <FILE-NAME> <DISK OUT DEFAULT OPTION>

<OUTPUT FILE
PARAMETERS> ::=          (<OUTPUT MEDIA> <RECORD SIZE>
<RECORDS PER BLOCK> <OUTPUT OPTION LIST>)

<OUTPUT MEDIA> ::=          CARD | CARDS | PAPER
| TAPE <PARITY PART>
| DISK (<RECORDS PER AREA> [<NUMBER OF AREAS>])
| PRINTER

<PARITY PART> ::=          (E) | (EVEN) | (O) | (ODD)
| <EMPTY>

<OUTPUT OPTION
LIST> ::=          <OUTPUT OPTION>
| <OUTPUT OPTION><OUTPUT OPTION LIST>

```

BURROUGHS CORPORATION
 COMPUTER SYSTEMS GROUP
 SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
 B1800/B1700 SORT LANGUAGE
 P.S. 2201 6752 REV. J

<OUTPUT OPTION> ::= <VARIABLE OPTION>
 | <MULTI OPTION>

<DISK OUT
 DEFAULT OPTION> ::= (DISK DEFAULT)

SEMANTICS

<FILE STATEMENT> contains information concerning the input and output files and records within the files.

<MULTIPLE INPUT FILE DESCRIPTIONS> specify that a merge or a multiple file disk sort is to be performed. Two to eight files may be merged and one to eight files may be sorted. The <DISK OUT DEFAULT OPTION> may not be used with merges.

If more than one file is input to the sort, the file numbers of the files to be sorted are put into the file table in a top down manner. The first eight bits of the table are set to zero for table format. The base relative address of the file table is stored in CT.2.

<INPUT FILE DESCRIPTION> consists of the word "IN" followed by <INPUT FILE PARAMETERS> or <DISK IN DEFAULT OPTION> for the input file. <OUTPUT FILE DESCRIPTION> consists of the word "OUT" followed by the <OUTPUT FILE PARAMETERS> or <DISK OUT DEFAULT OPTION> for the output file.

<INPUT FILE PARAMETERS> and <OUTPUT FILE PARAMETERS> contain the descriptive information required for sorting an input file and producing an output file. The <DISK IN DEFAULT OPTION> may be used when the input file is on disk. The descriptive information for the input file will be obtained from its disk file header, eliminating the need for parameter specifications. The <DISK OUT DEFAULT OPTION> may be used when the input file is on disk. The output file will then receive the same attributes as the input file (RECORD SIZE, BLOCKING FACTOR, AREAS, etc.). The parameters following the <FILE NAME> are enclosed within parentheses and are separated by a space.

<FILE-NAME> may consist of pack-id, multi-file-id, and file-id combined in accordance with standard B1800/B1700 file naming conventions.

BURROUGHS CORPORATION
COMPUTER SYSTEMS GROUP
SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
B1800/B1700 SORT LANGUAGE
P.S. 2201 6752 REV. J

<INPUT MEDIA> and <OUTPUT MEDIA> specify the hardware names for the input file and output file respectively. If the hardware type is DISK or PACK, the <RECORDS PER AREA> must be specified. If the hardware type is TAPE, the type of output tape may be optionally specified.

(<RECORDS PER AREA>) is an unsigned integer enclosed in parentheses specifying the number of records contained in a disk or pack area.

<RECORD SIZE> specifies the number of character positions the record will occupy in memory. If the record size is variable, <RECORD SIZE> is coded as the maximum number of positions which will be occupied by the largest record in the file. The maximum record length permitted is 65,535 bits. If the output record size and the input record size differ, the input record will be appropriately right truncated or right blank-filled to the output record size when moving it to output.

For blocked files having fixed length records, <RECORDS PER BLOCK> is an unsigned integer which specifies the number of logical records within a single block. <RECORDS PER BLOCK> is omitted for unblocked files.

The <PURGE OPTION> permits the user to optionally purge the input file after sorting by coding the reserved word "PURGE". If this entry is omitted, the input file is closed with release.

The <PARITY PART> option permits the user to specify the type of tape to receive the output file. If even or odd parity is specified, 7-track tape is assumed. If <PARITY PART> is omitted, 9-track, odd parity, either NRZ or PE tape is assumed. Tape density is not selectable.

The <VARIABLE OPTION> is used to specify that the input or output file consists of variable-length records. The V(<MAX BLOCK SIZE>) part is used for blocked variable-length records.

The <MULTI OPTION> is used with "PACK" or "DISK" to allow files to extend over more than one pack or cartridge.

BURROUGHS CORPORATION
COMPUTER SYSTEMS GROUP
SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
B1800/B1700 SORT LANGUAGE
P.S. 2201 6752 REV. J

The <OPTIONAL OPTIGN> allows the user to specify any input file as optional. If optional is specified and the file is not present, the user will be advised by a normal no file message from the MCP. The missing file can be loaded or an "OF" message entered to cause the file to be bypassed. The default option may not be used with "OPTIONAL."

BURROUGHS CORPORATION
 COMPUTER SYSTEMS GROUP
 SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
 B1800/B1700 SORT LANGUAGE
 P.S. 2201 6752 REV. J

SORT STATEMENT

SYNTAX

```

<SORT STATEMENT> ::=      <FIELD DESCRIPTION PART><BIAS PART>
                           <NUMBER OF RECORDS PART>

<FIELD DESCRIPTION
PART> ::=                  <FIELD PART><DESCRIPTION PART>

<FIELD PART> ::=          FIELD I FIELDS I KEY I KEYS

<DESCRIPTION PART> ::=    <DESCRIPTION PART><DESCRIPTION>
                           I <DESCRIPTION>

<DESCRIPTION> ::=        (<RELATIVE LOCATION PART><FIELD
                           LENGTH PART><SEQUENCE PART>
                           <FORMAT PART>)

<RELATIVE LOCATION
PART> ::=                 <UNSIGNED INTEGER>

<FIELD LENGTH
PART> ::=                 <UNSIGNED INTEGER>

<SEQUENCE PART> ::=      ASCENDING I A I DESCENDING
                           I D I EMPTY

<FORMAT PART> ::=        ALPHA I UA I NUMERIC I UN
                           I SA I SN I NC I RSA I RSN I <EMPTY>

<BIAS PART> ::=          BIAS <UNSSGND INTEGER 0 - 99>
                           <PERCENT PART> I EMPTY

<PERCENT PART> ::=      <PERCENT SIGN> I <EMPTY>

<NUMBER OF RECORDS
PART> ::=                 <UNSIGNED INTEGER> RECORDS
                           I EMPTY
  
```

BURROUGHS CORPORATION
COMPUTER SYSTEMS GROUP
SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
B1800/B1700 SORT LANGUAGE
P.S. 2201 6752 REV. J

SEMANTICS

<SORT STATEMENT>

The <SORT STATEMENT> contains control information which defines the sort key, fields within the key, and the sequence into which each field of the key is to be sorted. The description for each field is bounded by parentheses. The reserved word defined by the <FIELD PART> is coded once and applies to all multiple field descriptions. Fields to be sorted can be alphanumeric, signed alphanumeric, or unsigned four-bit numeric. All four formats are permitted in the same record in any order. Each individual field may be sorted to either ascending or descending sequence. The maximum number of sort keys is 30 unsigned keys, or 15 signed keys, or a combination of unsigned and signed keys. Here each signed key counts as two keys, the total not to exceed thirty.

<DESCRIPTION>

<DESCRIPTION> gives the location, length, sequence, and format of each field to be sorted. The location of the field is specified by an unsigned integer which indicates the character, digit, or sign position of the most significant character or digit of the field relative to the first character or digit in the record. The first character or digit of the record is considered to be location one. The length of the field is specified by an unsigned integer which indicates the number of character, or digit, positions contained within the field. The maximum size of the total key is 4,095 bits. The sequence to which the field is to be sorted is specified by the reserved words ASCENDING (or A) for ascending sequence and DESCENDING (or D) for descending sequence. If sequence is omitted, ascending is assumed. The format of the data in the field to be sorted may be specified by reserved words ALPHA (or UA) for unsigned alphanumeric, NUMERIC (or UN) for unsigned numeric, SA for signed alphanumeric (produces two keys), SN for signed numeric (produces two keys), RSA for right signed alpha (produces four keys), RSN for right signed numeric (produces two keys), or NC for unsigned alpha without collation. The sign for a RSA field is in the zone portion of the right most byte. The sign for a RSN field is in the right most digit. If format is omitted, alpha is assumed. The units of length and displacement are 8-bits for alpha or 4-bits for numeric. The elements of <DESCRIPTION> are separated by spaces.

BURROUGHS CORPORATION
COMPUTER SYSTEMS GROUP
SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
B1800/B1700 SORT LANGUAGE
P.S. 2201 6752 REV. J

<BIAS PART>

The information given by the <BIAS PART> is used to optimize the sort program. The value of the BIAS, in some cases, can be accurately estimated. If given, it is assumed to have been computed as follows:

$BIAS = 100K/(N-1)$, rounded to the nearest integer.

Here N is the estimated number of records in the file, and K is the estimated count of the occurrences of the relation:

$SORT\ KEY\ (I)th\ RECORD < OR = SORT\ KEY\ OF\ (I+1)st\ RECORD.$

Note that BIAS equals 100 for an already sorted file. If omitted, <BIAS PART> is assumed to be 50 percent, that is, completely random.

<NUMBER OF RECORDS>

The number of records is an estimate of the number of records in the file to be sorted. It is used in conjunction with the <BIAS PART> for optimization purposes. If omitted, <NUMBER OF RECORDS> is assumed to be 20,000. If the input is a disk file, the number of records is supplied by the disk file header.

USER OPTION LIST

SYNTAX

<USER OPTION STATEMENT> ::= <USER OPTION> | <USER OPTION>
<USER OPTION STATEMENT>

<USER OPTION> ::= <MEMORY SIZE OPTION> | <SORT TYPE
OPTION> | <SYNTAX OPTION>
| <SEQUENCE OPTION>
| <DISPLAY TIMING OPTION> | <ZIP OPTION>
| <RESTART OPTION> | <COLLATE OPTION>
| <NOPRINT OPTION> | <COMMENT>
| <PARITY ACTION> | <PARTITION OPTION>
| <DUPLICATE CHECKING OPTION>
| <TAGSORT OPTION> | <TAGSEARCH OPTION>
| <TAGINDEX OPTION>
| <INCLUDE-DELETE OPTION>
| <WORKPACKID OPTION> | <STABLE OPTION>

<MEMORY SIZE OPTION> ::= MEMORY <UNSIGNED INTEGER>
| <EMPTY>

<SORT TYPE OPTION> ::= DISKSORT | INPLACE | <NUMBER OF TAPES>
TAPESORT | MERGE | <EMPTY>

<SYNTAX OPTION> ::= SYNTAX | <EMPTY>

<SEQUENCE OPTION> ::= SEQUENCE | <EMPTY>

<DISPLAY TIMING OPTION> ::= TIMING | <EMPTY>

<ZIP OPTION> ::= ZIP (<ZIP STATEMENT>)
| <EMPTY>

<ZIP STATEMENT> ::= <ANY MCP CONTROL STATEMENT
OR CHARACTER>

<RESTART OPTION> ::= RESTART <JOB NUMBER>
| <EMPTY>

BURROUGHS CORPORATION
 COMPUTER SYSTEMS GROUP
 SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
 B1800/B1700 SORT LANGUAGE
 P.S. 2201 6752 REV. J

<JOB NUMBER> ::=	<UNSIGNED INTEGER>
<COLLATE OPTION> ::=	COLLATE (<FILE-NAME><COLLATE.TABLE.SPECS>)
<FILE-NAME> ::=	<ANY VALID FILE NAME (Quotes are required if blanks, quotes, slashes, left or right parentheses are used in a name)>
<COLLATE.TABLES.SPECS> ::=	(<LITERAL.LIST><SAVE.PART>) <EMPTY>
<SAVE.PART> ::=	SAVE <EMPTY>
<LITERAL.LIST> ::=	<LITERAL.PART><LITERAL.LIST> <LITERAL.PART>
<LITERAL.PART> ::=	<HEX.PART> <CHARACTER.PART>
<HEX.PART> ::=	@<HEX.DIGITS><RANGE.PART><HEX.DIGITS>@
<HEX.DIGITS> ::=	<ANY STRING OF PAIRS OF HEXADECIMAL CHARACTERS>
<RANGE.PART> ::=	- <EMPTY>
<CHARACTER.PART> ::=	"<CHAR.LIT>"
<CHAR.LIT> ::=	<ANY CHARACTERS EXCEPT QUOTE>
<NOPRINT OPTION> ::=	NOPRINT <EMPTY>
<COMMENT> ::=	<ANY NON-RESERVED WORD OR NON-RESERVED CHARACTER> EMPTY
<PARITY ACTION> ::=	PARITY DISCARD <EMPTY>
<PARTITION OPTION> ::=	PARTITION (<STARTING RECORD> <NUMBER OF RECORDS>) <EMPTY>
<STARTING RECORD> ::=	<UNSIGNED INTEGER GTR 0>
<NUMBER OF RECORDS> ::=	<UNSIGNED INTEGER GTR 0>
<DUPLICATE CHECKING OPTION> ::=	DUPCHECK DUPCHECK <IN-OR-OUT> <DUP-FUNCTION> <DUP-TYPE> <EMPTY>

BURROUGHS CORPORATION
 COMPUTER SYSTEMS GROUP
 SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
 B1800/B1700 SORT LANGUAGE
 P.S. 2201 6752 REV. J

<IN-OR-OUT> ::= IN I OUT

<DUP-TYPE> ::= ALL I FIRST I LAST

<DUP-FUNCTION> ::= LIST I COPY (<FILE-NAME>)
 I DUMP (<FILE-NAME>) I REMOVE

<TAGSORT OPTION> ::= TAGSORT I <EMPTY>

<TAGSEARCH OPTION> ::= TAGSEARCH I <EMPTY>

<TAGINDEX OPTION> ::= TAGRPG I TAGCOBOL I <EMPTY>

<INCLUDE-DELETE OPTION> ::= INCLUDE <IN-PART>
 <KEY.DESC.LIST> I DELETE <IN-PART>
 <KEY.DESC.LIST> I <EMPTY>

<IN-PART> ::= IN I <EMPTY>

<KEY.DESC.LIST> ::= <KEY.DESC> <LOGIC PART>
 I <KEY.DESC> <LOGIC PART>
 <KEY.DESC.LIST>

<KEY.DESC> ::= (<ITEM 1> <COMPARE TYPE> <ITEM 2>
 <LENGTH> <TYPE>)

<ITEM 1> ::= <INTEGER 1-8192 BYTES OR 1-16384 DIGITS>

<COMPARE TYPE> ::= EQL I NEQ I LSS I GTR I LEQ I GEQ
 I <EMPTY>

<ITEM 2> ::= "<1-3 CHARACTERS>" I @<1-6 HEX DIGITS>@
 I <INTEGER 1-8192 BYTES OR 1-16384 DIGITS>

<LENGTH> ::= <INTEGER 1-3> I <EMPTY>

<TYPE> ::= ALPHA I UA I NUMERIC I UN I <EMPTY>

<LOGIC PART> ::= OR I AND I <EMPTY (LAST MUST BE EMPTY)>

<WORKPACKID OPTION> ::= WORKPACK1 <NAME> I WORKPACK2 <NAME>
 I <EMPTY>

<NAME> ::= <1-10 CHARACTER DISK PACK OR
 CARTRIDGE NAME>

<STABLE OPTION> ::= STABLE I <EMPTY>

BURROUGHS CORPORATION
COMPUTER SYSTEMS GROUP
SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
B1800/B1700 SORT LANGUAGE
P.S. 2201 6752 REV. J

SEMANTICS

<USER OPTION STATEMENT>

Coding to exercise user option is via the <USER OPTION STATEMENT>. A space separates adjacent elements within the user options.

<MEMORY SIZE OPTION>

The amount of memory to be used by the SORT program is specified by the <MEMORY SIZE OPTION> and must be a minimum of 8,000 bytes. The option is coded by the reserved word MEMORY followed by an unsigned integer indicating the memory size in bytes. If <MEMORY SIZE OPTION> is omitted, 20,000 bytes is assumed.

Note: Maximum memory for an INPLACE SORT is 18,500 bytes.

Additional memory for all sorts will be dynamically assigned if:

- A. Available from roll-out of the calling program.
- B. Memory specified is insufficient for sorting.

Additional memory, up to 40K bytes, will be allocated in increments of 1K bytes.

<SORT TYPE OPTION>

<SORT TYPE OPTION> is used to specify one of four sort types (VECTOR RECORD SORT, INPLACE RECORD SORT, TAPESORT, or MERGE ONLY). If <SORT TYPE OPTION> is omitted or "DISKSORT" is entered, a VECTOR RECORD SORT is executed. INPLACE SORT ("INPLACE" entered) allows the user to execute a sort with a minimum of resources, and sort is executed without using additional disk space. <NUMBER OF TAPES> tapesort will cause a vector sort using tape for work files to be executed. <NUMBER OF TAPES> can be an integer from three to eight. If <SORT TYPE OPTION> is omitted and multiple input files are specified, a merge only is assumed. The keyword "MERGE" can also be used to specify a merge only.

<SYNTAX OPTION>

The final output desired is specified by the <SYNTAX OPTION>. If the <SYNTAX OPTION> is coded with the reserved word, SYNTAX, the input will be checked for syntactical errors, but the sort will not be executed. If <SYNTAX OPTION> is omitted, the sort intrinsic is executed.

<SEQUENCE OPTION>

The input file will be sequence checked, and the NUMBER OF RECORDS and BIAS will be printed on the SPO.

<DISPLAY TIMING OPTION>

This option causes generated sort parameters as a timing estimate of sort time to be printed.

<ZIP OPTION>

Allows the user to specify an operation to be performed by the MCP at the completion of a successful sort.

BURROUGHS CORPORATION
 COMPUTER SYSTEMS GROUP
 SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
 B1800/B1700 SORT LANGUAGE
 P.S. 2201 6752 REV. J

<RESTART OPTION>

The <RESTART OPTION> is used to restart an incompleted TAPESORT corresponding to the job number supplied in the <RESTART OPTION STATEMENT>.

<COLLATE OPTION>

SORT will create a virtual collating sequence file from its own specifications. The file will be created, locked on disk, used by the sorting program, and left on disk after the sort if SAVE is specified. The file will also be saved if compiling.

NOTE: If <COLLATE.TABLE.SPECS> part is <EMPTY>, the file specified by <FILE-NAME> will be accessed. The file specified must have been previously generated by sort by use of the <COLLATE OPTION> and <SAVE PART>. Use of files generated by other programs (i.e. CREATE/TABLE) can cause erroneous sort sequences or error conditions.

The table is created from the specified literals with the first item specified collating as hex 0. Any items not specified will collate as themselves. No item may be specified twice. Example:

```
COLLATE(COLLATE/1("9876543210"))
```

A collate file named COLLATE/1 will be created, used for the sort, and removed (unless an abnormal terminate occurs). The collating sequence will be "9876543210" @0A-C0@ "ABCDEFGHI" @CA-DC@ "JKLMNOPQR" @DA-E1@ "STUVWXYZ" @EA-EF@ @FA-FF@". The output will not contain characters of value @00-09@ as they were overlaid and not moved.

<NOPRINT OPTION>

Must be the first specification, if used. Nothing will be printed unless syntax errors are found. Sequence information will be printed on the SPO.

BURROUGHS CORPORATION
COMPUTER SYSTEMS GROUP
SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
B1800/B1700 SORT LANGUAGE
P.S. 2201 6752 REV. J

<COMMENT>

Comments may be interspersed among statements and options for documentation purposes. A colon, not in a ZIP, FILE-NAME, or quotes, terminates scanning of that card or record. Anything to the right of the colon will be ignored.

<PARITY ACTION>

Allows the user to specify whether records with parity errors should be discarded or cause termination of the sort. Default is termination. This option applies to DISKSORT, TAPESORT, or MERGE.

<PARTITION OPTION>

A specific portion of the input file may be sorted if this option is set. It applies only to VSORT or TAPESORT.

<DUPLICATE CHECKING OPTION>

When DUPCHECK is specified without parameters the last pass of the sort program will check for duplicate keys. If any are found a pointer file will be created and the file names will be displayed on the SPO. The pointer file would contain an eight byte record for each duplicate record. Each eight byte record would contain 2 four byte packed decimal relative pointers. The first would point to the duplicate record and the second would point to the record it duplicates.

When other parameters are specified then a single pass will be made over the file by SORT checking for duplicate keys. If "IN" is specified then a SORT will not be performed. The records will be checked after the SORT when "OUT" is specified.

The next parameter specifies whether the duplicate records are to be listed, copied to a second file, moved to a second file, or removed and discarded.

BURROUGHS CORPORATION
 COMPUTER SYSTEMS GROUP
 SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
 B1800/B1700 SORT LANGUAGE
 P.S. 2201 6752 REV. J

The last parameter determines the duplicate record grouping that will be used by SORT. All but the first is specified by "FIRST" and all but the last is specified by "LAST". deleted. deleted. If nothing else is specified then the duplicate records will be checked for by the sort program during the last sorting pass. In this case a file of duplicate record pointers will be created. The file will be called SD.<JOB NUMBER> and will contain 8 character records which are two 4 character pack decimal pointers to the duplicate records and to the records they are duplicates of, respectively. If the <IN-OR-OUT> part is specified then the function will be handled by SORT. The <IN-OR-OUT> part determines whether or not the input file is to be searched. Duplicate records may be listed, copied into a second file, removed to a second file, or removed and discarded. Three groups of duplicate records may be considered, "FIRST", "LAST", and "ALL". The "FIRST" and "LAST" group are all duplicates except the first and last respectively. For example, to "DUPCHECK IN LIST FIRST" and there were three records the same in the input, then the last two would be printed.

<TAGSORT OPTION> <TAGSEARCH OPTION> <TAGINDEX OPTION>
 (TAGRPG | TAGCOBOL)

When memory and/or disk space are restricted, the use of these options may improve sorting time. The specification of any one of them causes the sort to build workfile records consisting only of the keys to be sorted on, and a 24-bit relative record pointer.

In most cases this is considerably smaller than the input or output records and will, therefore, sort faster. The output file will be different for each of these options.

TAGSORT will produce a file of four byte, packed decimal relative record pointers only. TAGSEARCH will use the pointers to randomly read the input records (disk only) and output them in sorted order. TAGRPG will produce a file of three byte, packed decimal pointers and the specified keys. TAGCOBOL will do the same except that the pointers will be four bytes. TAGSORT, TAGRPG, or TAGCOBOL may not output to cards or to the printer.

<INCLUDE-DELETE OPTION>

A total of 10 keys will be allowed. Each key may compare up to three characters. AND and OR logic will be provided for continuing from one key to the next. Each key may compare a

BURROUGHS CORPORATION
COMPUTER SYSTEMS GROUP
SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
B1800/B1700 SORT LANGUAGE
P.S. 2201 6752 REV. J

field in the record to either a constant contained in the key or to another field in the record. The displacement values become 0-8191 for byte and 0-16383 for digit when stored in the keys. The quotes delimit character literals which may contain any three characters except the quote and question mark. The signs denote hex literals and may contain only 1-6 hex digits. Character literals are left justified with blank fill and truncation on the right. Hex literals are right justified with zero fill and truncation on the left. All literals are right justified in the KEY.DATA field. SORT will set up the key table from the above information. If IN is specified, then SORT will perform the function; otherwise, the proper Sort Intrinsic will be called. The reserved words, INCLUDE and DELETE, may be used only once, but each may be followed by several keys. The S-op that compares these keys terminates on either a zero compare type or the length of the table.

<WORKPACKID OPTION>

Pack ID's may be specified so the DISK SORT'S workfiles can be separated on one or more disk drives. This speeds the sorting by cutting down read-write head contention.

<STABLE OPTION>

This option may be specified for SORT/VSORT and SORT/TAPESORT. The use of this option causes duplicate records to retain their original order. Default is not stable.

BURROUGHS CORPORATION
COMPUTER SYSTEMS GROUP
SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
B1800/B1700 SORT LANGUAGE
P.S. 2201 6752 REV. J

FILE NAMES

The current release of SORT is consistent with the MCP's file naming conventions except that quotes are still required around names containing special SORT characters such as left parenthesis, right parenthesis, and quote. As restricted by the MCP, blanks, slashes and quotes cannot appear in filenames unless they are quoted. The special characters enumerated are used as delimiters for SORT or MCP statements, and are therefore prohibited in filenames.

Users who run under the file security mechanism available in the current MCP should take special note of the parentheses required in the multifile-ids of secure files. These multifile-ids must be quoted, e.g., PACKA/"(USER)"/FILE, since both the left and right parenthesis occur in the filename. If the quotes are not included, SORT produces an error message saying that it cannot locate an input or output file.

BURROUGHS CORPORATION
 COMPUTER SYSTEMS GROUP
 SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
 B1800/B1700 SORT LANGUAGE
 P.S. 2201 6752 REV. J

SORT COMMUNICATE

CT.VERB = 29

CT.OBJECT = Base relative address of SORT Information Table

CT.ADVERB = SORT.RESTART BIT(1) %IF POSSIBLE
 SORT.DUPCHECK BIT(1) %CHECK FOR DUPS
 SORT.W1.PID BIT(1) %PACK ID, 0 = SYSTEM
 SORT.W2.PID BIT(1) %PACK ID, 0 = SYSTEM
 MUL_INPUT_FLAG BIT(1) %0=CT.2 is FPB NUMBER
 RESERVED BIT(7) %1=CT.2 is ADDRESS

CT.1 = Base relative address of SORT Key Table

CT.2 = If MERGE or MULTIPLE FILE INPUT, then base relative address of FILE TABLE, else input file number

CT.3 = If QSORT.SAME.NAME.IN.OUT then input file number, else output file number

CT.4 = Collate file number or not zero

CT.5 = Zero

CT.6 = Base relative address of the Delete Key Table or not zero.

CT.7 = If SORT.W1.PID then base relative address of 10 character name, else not zero.

CT.8 = If SORT.W2.PID then base relative address of 10 character name, else not zero.

REINSTATE STATUS

After a normal EOJ of the sorting program, the reinstate message pointer of the caller will contain the SORTs actual job number in binary.

BURROUGHS CORPORATION
COMPUTER SYSTEMS GROUP
SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
B1800/B1700 SORT LANGUAGE
P.S. 2201 6752 REV. J

SORT EOJ COMMUNICATE

CT.VERB = 33
CT.OBJECT = File number to be closed
CT.ADVERB = CLOSE type
CT.1 = EOF pointer
CT.2 = Record size

BURROUGHS CORPORATION
 COMPUTER SYSTEMS GROUP
 SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
 B1800/B1700 SORT LANGUAGE
 P.S. 2201 6752 REV. J

SORT INFORMATION TABLE FORMAT

02	SORT.TYPE	BIT (02), %00 = VECTOR, 01 = INPLACE, %10 = TAG, 11 = MERGE
02	SORT.HDWR	BIT (06), %MAY BE DISK, OR TAPE %(CODED AS FPB.HDWR FOR MCP)
02	FILLER	BIT (08)
02	SORT.PURGE.IN- PUT.FILES	BIT (08) %1 = CLOSE CORRESPONDING INPUT %FILE WITH PURGE
02	SORT.FILES	BIT (08), %NUMBER OF EU'S OR TAPES
02	SORT.RECSIZE	BIT (24), %LARGEST KEY DISPLACEMENT
02	SORT.IN.HDWR	BIT (06), %HARDWARE TYPE
02	SORT.IN.RECSIZE	BIT (24), %MAXIMUM RECORD SIZE IN BITS
02	SORT.IN.BLKSIZE	BIT (24), %MAXIMUM BLOCK SIZE IN BITS
02	SORT.IN.CLOSE	BIT (12), %CLOSE TYPE
02	SORT.IN.VARIABLE	BIT (01), %1 = VARIABLE RECORDS
02	SORT.OUT.HDWR	BIT (06), %HARDWARE TYPE
02	SORT.OUT.RECSIZE	BIT (24), %MAXIMUM RECORD SIZE IN BITS
02	SORT.OUT.BLKSIZE	BIT (24), %MAXIMUM BLOCK SIZE IN BITS
02	SORT.OUT.VARIABLE	BIT (01), %1 = VARIABLE RECORDS
02	SORT.OUT.CLOSE	BIT (12), %CLOSE TYPE
02	SORT.DELETING	BIT (01), %INCLUDE OR DELETE RECORDS
02	SORT.STABILIZE	BIT (01), %0 = NOT STABLE
02	SORT.PARITY	BIT (01), %0 = DS, 1 = IGNORE RECORD
02	SORT.RESTART	BIT (01) %1=RESTART
02	SORT.BIAS	BIT (07), %RANGE 0-99 %DEFAULT = 50 (RANDOM)
02	SORT.RECORDS	BIT (24), %# OF RECORDS IN INPUT FILE %DEFAULT = 20,000
02	SORT.TIMING	BIT (01), %0 = NO REPORT WANTED %1 = PRINT SORT SPECS
02	SORT.NUMBER.KEYS	BIT (05), %NUMBER OF SORT KEYS
02	FILLER	BIT (08), %
02	SORT.KEY.LENGTH	BIT (16), %TOTAL LENGTH OF KEY(S) %IN BITS
02	FILLER	BIT (16), %0
02	SORT.PARTITION	BIT (24), %PARTITION STARTING RECORD
02	SORT.NUMBER.DELETE.KEYS	BIT (04), %0-10 DELETE KEYS
02	SORT.DUPCHECK	BIT (01), %
02	SORT.TAGRPG	BIT (01), %3 BYTE TAGS
02	SORT.W1.PID	BIT (01), %WORKFILE W1 PACK ID
02	SORT.W2.PID	BIT (01), %WORKFILE W2 PACK ID
02	SORT.TAGCOBOL	BIT (01), %4 BYTE TAGS
02	FILLER	BIT (15), %0
02	SORT.MEMORY	BIT (24), %MEMORY SIZE IN BITS
02	SORT.TAGSEARCH	BIT (01), %TAGSEARCH
02	SORT.COLLATE	BIT (01), %COLLATE
02	FILLER	BIT (31), %RESERVED FOR DEBUG
02	SORT.RESTART.JOB	BIT (24), %JOB NUMBER OF SORT %BEING RESTARTED

BURROUGHS CORPORATION
 COMPUTER SYSTEMS GROUP
 SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
 B1800/B1700 SORT LANGUAGE
 P.S. 2201 6752 REV. J

SORT KEY TABLE

The SORT Key Table may contain a maximum of thirty (30) 36-bit Key descriptions. Each description will specify ascending or descending order, key length, and key displacement. Additional description must be supplied for signed keys which specifies the type, length, and displacement of the sign.

FORMAT:

01	SORT.KEY.TABLE	BIT(1116),	%MAX 30 KEYS+ZERO KEY(ONE)
	%UNSIGNED KEY(S)		
	02 KEY.FLAGS	BIT(04),	
	03 SIGN.FLAG	BIT(01),	%0 = UNSIGNED
	03 DIRECTION	BIT(01),	%0 = ASCENDING %1 = DESCENDING
	03 FILLER	BIT(01),	%=0
	03 TRAN.KEY	BIT(01),	%COLLATE OPTION (UNSIGNED %ALPHA KEYS ONLY)
	02 KEY.LENGTH	BIT(12),	%LENGTH OF KEY IN BITS
	02 KEY.DISPLACEMENT	BIT(20),	%LOCATION FROM BEGINNING %OF RECORD
	%SIGNED KEYS		
	02 KEY.FLAGS	BIT(04),	
	03 SIGN.FLAG	BIT(01),	%1 = SIGNED
	03 FILLER	BIT(01),	%= 0
	03 NEW.FORMAT	BIT(01),	%= 1
	03 FILLER	BIT(01),	%= 0
	02 LENGTH.AND.SIGN	BIT(12),	%= 240D2
	03 LENGTH	BIT(04),	%SIGN LENGTH
	03 SIGN	BIT(08),	%SIGN
	02 SIGN DISPLACEMENT	BIT(20),	%LOCATION OF SIGN FROM %BEGINNING OF RECORD

BURROUGHS CORPORATION
 COMPUTER SYSTEMS GROUP
 SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
 B1800/B1700 SORT LANGUAGE
 P.S. 2201 6752 REV. J

MERGE/MULTIPLE INPUT FILE TABLE

01	MUL_INPUT_TABLE	BIT(80),
02	FILLER	BIT(08), % = 0
02	MERGE.DISK.IN	BIT(08), % NUMBER OF INPUT DISK FILES
02	MRGE.INPT.FILE(8)	BIT(08), % 2-8 INPUT FILE NUMBERS

DELETE KEY TABLE

01	DELETE.KEYS (10)	BIT(48)
02	KEY.TYPE	BIT(1) % 0 = INC 1 = DEL
02	LOGIC.TYPE	BIT(1) % 0 = OR 1 = AND
02	DATA.TYPE	BIT(1) % 0 = DATA 1 = FIELD DESC.
02	LENGTH.TYPE	BIT(1) % 0 = DIGIT 1 = BYTE
02	COMPARE.TYPE	BIT(4)
		% GTR = 1 LSS = 2 NEQ = 3
		% EQL = 4 GEQ = 5 LEQ = 6
02	KEY.LENGTH	BIT(2)
02	KEY.DISPL.1	BIT(14) % ITEM 1
02	KEY.DATA	BIT(24) % ITEM 2
	03 FILLER	BIT(10)
	03 KEY.DISPL.2	BIT(14)

BURROUGHS CORPORATION
 COMPUTER SYSTEMS GROUP
 SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
 B1800/B1700 SORT LANGUAGE
 P.S. 2201 6752 REV. J

SORT EXECUTION

SORT can be executed from cards, from the SPO (console keyboard), or from a remote terminal. If SW0=1, SORT specifications are accepted from the SPO (or remote) instead of the CARDS file. If Program Switch One is set (SW1=1), errors and their source text are written to a printer or a backup print file named ERROR.LINE. This option is useful to datacomm users who wish to see any errors displayed on their terminals. The job number of the actual sorting program is returned to the calling program in its RS.REINSTATE.MSG.PTR. It is a 24-bit binary value.

SAMPLE EXECUTIONS

For a single SORT, the following execute deck would be used:

<u>CARD</u>	<u>COLUMN #1</u>	<u>COLUMN #2-80</u>
----	-----	-----
#1	?	EXECUTE) SORT
#2	?	DACTA) CARDS
		<STATEMENT CARDS>
LAST	?	END

For a SORT that can be used over and over, a compile deck would be used:

<u>CARD</u>	<u>COLUMN #1</u>	<u>COLUMN #2-80</u>
----	-----	-----
#1	?	CO(MPILE) <GIVEN NAME>
		(WITH) SORT LI(BRARY)

BURROUGHS CORPORATION
COMPUTER SYSTEMS GROUP
SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
B1800/B1700 SORT LANGUAGE
P.S. 2201 6752 REV. J

To run a SORT from the SPO, use:

```
EXECUTE) SORT SW 0 = 1;  
OR  
CO(MPILE) <GIVEN NAME> (WITH) SORT LI(BRARY) SW 0 = 1;
```

The SPO will respond with:

```
<BOJ MESSAGE>  
SORT = <JOB NUMBER> ACCEPT
```

Enter specifications by:

```
<JOB NUMBER> AX <SORT STATEMENTS>  
.  
.  
<JOB NUMBER> AX <BLANK>
```

The blank line terminates the input phase.

To run a SORT from a remote terminal and send errors to printer backup, use:

```
EXECUTE) SORT SW0=1 SW1=1;  
or  
CO(MPILE) <GIVEN NAME> (WITH) SORT LI(BRARY) SW0=1 SW1=1;
```

BURROUGHS CORPORATION
COMPUTER SYSTEMS GROUP
SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
B1800/B1700 SORT LANGUAGE
P.S. 2201 6752 REV. J

ERROR MESSAGES

The following error messages are provided by SORT:

1. "INVALID OR MISSING RECORDS PER AREA"
2. "INVALID OR MISSING DEVICE TYPE"
3. "INVALID OR MISSING RECORD SIZE"
4. "INVALID OR MISSING NAME IN PARENTHESES"
5. "INVALID OR MISSING FILE STATEMENT"
6. "INVALID OR MISSING VARIABLE BLOCK SIZE"
7. "INVALID OR MISSING FILE NAME"
8. "INVALID OR MISSING KEY LOCATION"
9. "INVALID OR MISSING KEY LENGTH"
10. "INVALID OR MISSING KEY STATEMENT"
11. "INVALID OR MISSING RIGHT PARENTHESIS"
12. "INVALID OR MISSING LITERAL"
13. "INVALID OR MISSING COLLATE STATEMENT"
14. "INVALID OR MISSING DELIMITER"
15. "PARITY IS (EVEN) OR (ODD)"
16. "ZIP IS (255 CHARACTERS IN DELIMITERS)"
17. "PARAMS ARE (STARTING.RECORD HOW.MANY)"
18. "BIAS IS C-99"
19. "MEMORY RANGE IS 8000 TO 125000 BYTES"
20. "NO DISK DEFAULT OUT FOR MERGE"
21. "DEFAULT ONLY WITH DISK"
22. "LENGTH 1 SIGNED NUMERIC KEY REQUIRES UNSIGNED KEY AFTER"
23. "NUMBER IS OUT OF RANGE"
24. "TOO MANY FILE IN OR OUT PARTS"
25. "CANNOT CLOSE OUTPUT DISK RELEASE"
26. "TOO MANY KEYS SPECIFIED"
27. "SORT KEY OUT OF RECORD"
28. "PURGE INPUT ONLY"
29. "8 BYTES MIN RECORD FOR TAPE"
30. "MULTI ON DISK ONLY"
31. "RECORDS PER AREA NOT EVEN MULTIPLE OF BLOCKS PER AREA"
32. "DISK FILE MUST BE LABELED"
33. "TAPESORT NOT PRECEDED BY NUMBER"
34. "RECORDS OPTION NOT PRECEDED BY NUMBER"
35. "MAX MEMORY FOR INPLACE SORT IS 18,500"
36. "VARIABLE LENGTH RECORDS NOT ALLOWED"
37. "TAGSEARCH MUST HAVE DISK INPUT"
38. "TAGSORT MUST HAVE OUTPUT RECORD SIZE = 4 BYTES"
39. "INPLACE SORT CANNOT HANDLE TAGSORT"
40. "VARIABLE OUT MUST HAVE VARIABLE IN"
41. "NO INPLACE TRANSLATE"
42. "TAPESORT NOT AN INPLACE SORT"

BURROUGHS CORPORATION
COMPUTER SYSTEMS GROUP
SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
B1800/B1700 SORT LANGUAGE
P.S. 2201 6752 REV. J

43. "INVALID INCLUDE DELETE KEY"
44. "RECORDS PER BLOCK CANNOT BE 0"
45. "DELETE KEY OUT OF RECORD"
46. "DEFAULT MUST BE DISK INPUT"
47. "CANNOT COMPILE THESE OPTIONS"
48. "DUPCHECK IN/OUT LIST/COPY(N)/DUMP(N)/REMOVE ALL/FIRST/LAST"
49. "INVALID USE OF LEFT PARENTHESIS"
50. "TAGSORT OUTPUT CANNOT BE CARD, PRINT, OR TAPE"
51. "SPECIFIED MORE THAN ONCE"
52. "INPLACE SORT CANNOT DELETE RECORDS OR PARTITION"
53. "INVALID TAGSORT OPTION COMBINATION"
54. "TAGRPG OUTPUT RECORD SIZE = 3 BYTES + KEYS"
55. "PARAMS ARE DV DRPT MRPT BLBUMP FLAG"
56. "CANNOT DO THESE OUTPUT OPTIONS" % 55
57. "TAGCOBOL OUTPUT RECORD SIZE = 4 BYTES + KEYS"
58. "MISSING LEFT PARENTHESIS"
59. "INPLACE SORT NOT STABLE"
60. "INPLACE SORT CANNOT PARTITION"
61. "OPTIONAL FILE AND DEFAULT NOT ALLOWED"

BURROUGHS CORPORATION
COMPUTER SYSTEMS GROUP
SANTA BARBARA PLANT

COMPANY CONFIDENTIAL
B1800/B1700 SORT LANGUAGE
P.S. 2201 6752 REV. J

INDEX

ERROR MESSAGES 10-1
FILE NAMES 4-10
FILE STATEMENT 2-1
GENERAL 1-1
REINSTATE STATUS 5-1
RELATED DOCUMENTATION 1-2
SAMPLE EXECUTIONS 9-1
SEMANTICS 2-3, 3-2, 4-4
SORT COMMUNICATE 5-1
SORT EOJ COMMUNICATE 6-1
SORT EXECUTION 9-1
SORT INFORMATION TABLE FORMAT 7-1
SORT KEY TABLE 8-1
SORT STATEMENT 3-1
SYNTAX 2-1, 3-1, 4-1
USER OPTION LIST 4-1